Maple-assisted proof of formula for A295602

Robert Israel

24 November 2017

There are $2^8 = 256$ possible configurations for a 2×4 sub-array. Consider the 256×256 transition matrix T such that $T_{ij} = 1$ if the bottom two rows of a 3×4 sub-array could be in configuration i while the top two rows are in configuration j (i.e. the middle row is compatible with both i and j, and each 1 in that row is horizontally or vertically adjacent to 0,2, 3 or 4 1's), and 0 otherwise. The following Maple code computes it. I'm encoding a configuration

$$\begin{bmatrix} b_1 & b_2 & b_3 & b_4 \\ b_5 & b_6 & b_7 & b_8 \end{bmatrix}$$

as $b+1$ where $b_1 b_2 b_3 b_4 b_5 b_6 b_7 b_8$ is the binary representation of b. The $+1$ is needed because matrix indices start at 1 rather than 0.

```
> q:= proc(a,b) local r,s,t,M,i;
     s:= floor((a-1)/16);
     if s <> (b-1) mod 16 then return 0 fi;
     s:= convert(s+16,base,2);
     r:= convert(16+floor((b-1)/16),base,2);
     t:= convert(16+ ((a-1) mod 16),base,2);
     M:= Vector(4);
     for i from 1 to 3 do if s[i] = 1 and s[i+1] = 1 then M[i]:= M
  [i]+1; M[i+1]:= 1 fi od;
     for i from 1 to 4 do if s[i]=1 then
       M[i]:= M[i]+r[i]+t[i];
       if M[i] =1 then return 0 fi;
     fi od;
     1
  end proc:
  T:= Matrix(256,256, q);
```

$$T := \begin{bmatrix} \text{256 x 256 Matrix} \\ \text{Data Type: anything} \\ \text{Storage: rectangular} \\ \text{Order: Fortran\_order} \end{bmatrix} \qquad \textbf{(1)}$$

Thus $a(n) = u\, T^n\, v$ where u and v are row and column vectors respectively with $u_i = 1$ for i corresponding to configurations with bottom row $(0, 0, 0, 0)$, 0 otherwise, and $v_i = 1$ for i corresponding to configurations with top row $(0, 0, 0, 0)$, 0 otherwise. The following Maple code produces these vectors.

```
> u:= Vector[row](256):
  v:= Vector(256):
  for i from 0 to 15 do u[16*i+1]:= 1; v[i+1]:= 1;
  od:
```

To check, here are the first few entries of our sequence.

```
> seq(u . T^n . v, n = 1 .. 10);
```

$$8, 51, 338, 2305, 16340, 119371, 892086, 6775059, 52046892, 402986355 \quad \textbf{(2)}$$

Now here is the minimal polynomial P of T, as computed by Maple.

```
> P:= unapply(LinearAlgebra:-MinimalPolynomial(T, t), t);
```

$$P := t \to t^{51} - 16\,t^{50} + 85\,t^{49} - 191\,t^{48} + 180\,t^{47} + 294\,t^{46} - 1545\,t^{45} + 5544\,t^{44} - 9860\,t^{43} + 23597\,t^{42} - 19388\,t^{41} + 62109\,t^{40} - 60367\,t^{39} - 396253\,t^{38} + 351775\,t^{37} - 2701946\,t^{36} - 521060\,t^{35} - 2826153\,t^{34} - 7988033\,t^{33} + 513081\,t^{32} - 11139109\,t^{31} - 17433782\,t^{30} - 27265572\,t^{29} - 3441067\,t^{28} - 78717856\,t^{27} - 1078113\,t^{26} + 77143348\,t^{25} - 6546049\,t^{24} + 187802715\,t^{23} + 85709095\,t^{22} - 210420921\,t^{21} - 73223436\,t^{20} + 63913788\,t^{19} + 17972703\,t^{18} + 5601845\,t^{17} - 4355133\,t^{16} - 17598342\,t^{15} + 2153086\,t^{14} + 10098595\,t^{13} - 1314498\,t^{12} - 3258038\,t^{11} - 39951\,t^{10} + 300206\,t^{9} + 366501\,t^{8} + 38499\,t^{7} - 35235\,t^{6} - 7817\,t^{5} - 1772\,t^{4} - 237\,t^{3} + 23\,t^{2} \quad \textbf{(3)}$$

This turns out to have degree 51. Thus we will have $0 = u\,P(T)\,T^n\,v = \sum_{i=0}^{51} p_i\,a(i+n)$ where p_i is the coefficient of t^i in $P(t)$. That corresponds to a homogeneous linear recurrence of order 51, which would hold true for any u and v. It seems that with our particular u and v we have a recurrence of order only 27, corresponding to a factor of P.

```
> factor(P(t));
```

$$t^2\,(t^{27} - 13\,t^{26} + 43\,t^{25} - 33\,t^{24} + 97\,t^{23} - 73\,t^{22} + 730\,t^{21} - 503\,t^{20} - 2269\,t^{19} - 2715\,t^{18} - 15348\,t^{17} - 11653\,t^{16} - 23118\,t^{15} - 14969\,t^{14} - 12525\,t^{13} + 73129\,t^{12} + 39381\,t^{11} - 23137\,t^{10} + 1963\,t^{9} - 10808\,t^{8} - 479\,t^{7} - 1751\,t^{6} - 102\,t^{5} - 5301\,t^{4} + 770\,t^{3} + 368\,t^{2} + 62\,t + 23)\,(t^{22} - 3\,t^{21} + 3\,t^{20} + 10\,t^{19} - 15\,t^{18} + 132\,t^{17} - 94\,t^{16} + 93\,t^{15} + 502\,t^{14} - 760\,t^{13} + 959\,t^{12} + 478\,t^{11} - 41\,t^{10} + 2338\,t^{9} - 102\,t^{8} - 1881\,t^{7} + 66\,t^{6} + 536\,t^{5} + 86\,t^{4} - 9\,t^{3} - 58\,t^{2} - 13\,t + 1) \quad \textbf{(4)}$$

```
> Q:= unapply(t^27-13*t^26+43*t^25-33*t^24+97*t^23-73*t^22+730*t^21
  -503*t^20-2269*t^19-2715*t^18-15348*t^17-11653*t^16-23118*t^15
  -14969*t^14-12525*t^13+73129*t^12+39381*t^11-23137*t^10+1963*t^9
  -10808*t^8-479*t^7-1751*t^6-102*t^5-5301*t^4+770*t^3+368*t^2+62*
  t+23, t);
```

$$Q := t \to 23 + 62\,t + t^{27} - 13\,t^{26} + 43\,t^{25} - 33\,t^{24} + 97\,t^{23} - 73\,t^{22} + 730\,t^{21} - 503\,t^{20} - 2269\,t^{19} - 2715\,t^{18} - 15348\,t^{17} - 11653\,t^{16} - 23118\,t^{15} - 14969\,t^{14} - 12525\,t^{13} + 73129\,t^{12} + 39381\,t^{11} - 23137\,t^{10} + 1963\,t^{9} - 10808\,t^{8} - 479\,t^{7} - 1751\,t^{6} - 102\,t^{5} - 5301\,t^{4} + 770\,t^{3} + 368\,t^{2} \quad \textbf{(5)}$$

The complementary factor $R(t) = \dfrac{P(t)}{Q(t)}$ has degree 24.

```
> R:= unapply(normal(P(t)/Q(t)), t);
```

$$R := t \to (t^{22} - 3\,t^{21} + 3\,t^{20} + 10\,t^{19} - 15\,t^{18} + 132\,t^{17} - 94\,t^{16} + 93\,t^{15} + 502\,t^{14} - 760\,t^{13} + 959\,t^{12} + 478\,t^{11} - 41\,t^{10} + 2338\,t^{9} - 102\,t^{8} - 1881\,t^{7} + 66\,t^{6} + 536\,t^{5} + 86\,t^{4} - 9\,t^{3} - 58\,t^{2} - 13\,t + 1)\,t^2 \quad \textbf{(6)}$$

Now we want to show that $b(n) = u\,Q(T)\,T^n\,v = 0$ for all n. This will certainly satisfy the order-24 recurrence

$$\sum_{i=0}^{24} r_i\, b(i+n) = \sum_{i=0}^{24} r_i u\, Q(T)\, T^{n+i} v = u\, Q(T)\, R(T)\, T^n\, v = u\, P(T)\, T^n\, v = 0$$

where r_i are the coefficients of $R(t)$. To show all $b(n) = 0$ it suffices to show $b(0) = \ldots = b(23) = 0$. This would take some time to do naively, so it's worthwhile to do some pre-calculation.

```
> w:= u . Q(T):
  V[0]:= v:
  for i from 1 to 23 do V[i]:= T . V[i-1] od:
  seq(w . V[i],i=0..23);
```

0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0 **(7)**